' =========================================================================

'

'   File...... ADC MCP3202 demo.bs2

'   Purpose... MCP3202 Demo

'   Author.... Technical Support -- Parallax, Inc.

'   E-mail.... support@parallax.com

'   Started... 8/20/09

'   Updated... N/A

'

'   {$STAMP BS2}

'   {$PBASIC 2.5}

'

' =========================================================================

' -----[ Program Description ]---------------------------------------------

'  This demo uses the wipers on 2 10K Ohm pots to provide an analog voltage

' for channel 0 and 1 on the ADC; then displays the voltage and 12-bit value

' for each channel on the DEBUG terminal. Using a 5 VDC reference the 12-bit

' values should be roughly 0 TO 4096 and the voltage from 0 to 5

' -----[ Revision History ]------------------------------------------------

  ' N/A

' -----[ I/O Definitions ]-------------------------------------------------

CS              PIN     0                         ' Chip Select  (MCP3202.1)

Clock           PIN     1                         ' Clock        (MCP3202.7)

DataOut         PIN     2                         ' Pin 6 on ADC (MCP3202.6)

DataIn          PIN     3                         ' Pin 5 on ADC (MCP3202.5)

' -----[ Constants ]-------------------------------------------------------

Cnts2Mv         CON     $0139                     ' x 1.22 (To Millivolts)

' -----[ Variables ]-------------------------------------------------------

result0         VAR     Word                      ' Conversion Result CH0

result1         VAR     Word                      ' Conversion Result CH1

mVolts0         VAR     Word                      ' Result0 --> mVolts

mVolts1         VAR     Word                      ' Result1 --> mVolts

' -----[ Init Setup ]------------------------------------------------------

DEBUG CLS, "ADC CH 0 : ", CR, "Volts    :", CR,

           "ADC CH 1 : ", CR, "Volts    :"

' -----[ Program Code ]----------------------------------------------------

DO

  LOW CS                                          ' Enable ADC

  SHIFTOUT DataIn, Clock, MSBFIRST, [%1101\4]     ' Select CH0, Single-Ended

  SHIFTIN DataOut, Clock, MSBPOST, [result0\12]   ' Read ADC

  HIGH CS                                         ' Disable ADC

  mVolts0 = result0 */ Cnts2Mv                    ' Convert To Millivolts

  LOW CS                                          ' Enable ADC

  SHIFTOUT DataIn, Clock, MSBFIRST, [%1111\4]     ' Select CH1, Single-Ended

  SHIFTIN DataOut, Clock, MSBPOST, [result1\12]   ' Read ADC

  HIGH CS                                         ' Disable ADC

  mVolts1 = result1 */ Cnts2Mv                    ' Convert To Millivolts

  DEBUG HOME, CRSRXY, 11, 0, DEC result0, CLREOL, ' Displays voltages & digital value

              CRSRXY, 11, 1, DEC mVolts0 DIG 3,   ' for both channels

                       ".", DEC3 mVolts0,

              CRSRXY, 11, 2, DEC result1, CLREOL,

              CRSRXY, 11, 3, DEC mVolts1 DIG 3,

                       ".", DEC3 mVolts1

  PAUSE 100

LOOP
